# 设定源文件文件夹
aux_source_directory(lib/ NETCDF_SRC)

find_package(HDF5 REQUIRED)

# 寻找netCDF库地址
find_package(netCDF REQUIRED)
if(netCDF_FOUND)
    include_directories(${netCDF_INCLUDE_DIR})
endif()

# 设置编译选项
set(CMAKE_C_FLAGS "${CMAKE_C_FLAGS} -O3")
set(CMAKE_CXX_FLAGS "${CMAKE_CXX_FLAGS} -O3")
# 设置库文件的输出地址
set(LIBRARY_OUTPUT_PATH ${PROJECT_BINARY_DIR}/lib)

# 以下部分为库的编译
# 注意目标名必须唯一 所以不能直接生成相同名称的动态库与静态库
# 注意此处不必为目标名称添加lib前缀和相应后缀，cmake会自行添加
add_library(netcdfcxx_legacy SHARED ${NETCDF_SRC})
# 首先添加静态库的生成命令
add_library(netcdfcxx_legacy_static STATIC ${NETCDF_SRC})
# 设置静态库的输出名称从而获得与动态库名称相同的静态库
set_target_properties(netcdfcxx_legacy_static PROPERTIES OUTPUT_NAME "netcdfcxx_legacy")
# 设置输出目标属性以同时输出动态库与静态库
set_target_properties(netcdfcxx_legacy PROPERTIES CLEAN_DIRECT_OUTPUT 1)
set_target_properties(netcdfcxx_legacy_static PROPERTIES CLEAN_DIRECT_OUTPUT 1)
# 设置动态库的版本号
set_target_properties(netcdfcxx_legacy PROPERTIES VERSION ${PROJECT_VERSION} SOVERSION ${PROJECT_VERSION_MAJOR}.${PROJECT_VERSION_MINOR})
# 设置动态库的运行搜索地址
set_target_properties(netcdfcxx_legacy PROPERTIES INSTALL_RPATH /usr/local/lib)
set_target_properties(netcdfcxx_legacy_static PROPERTIES INSTALL_RPATH /usr/local/lib)

#find_library(NETCDF_LIBRARY netcdf ${netCDF_LIB_DIR})
target_link_libraries(netcdfcxx_legacy PUBLIC ${netCDF_LIBRARIES})
target_link_libraries(netcdfcxx_legacy_static ${netCDF_LIBRARIES})
target_link_libraries(netcdfcxx_legacy PUBLIC ${HDF5_LIBRARIES})
target_link_libraries(netcdfcxx_legacy_static ${HDF5_LIBRARIES})

# 设置库文件的输出地址
set(LIBRARY_OUTPUT_PATH ${PROJECT_BINARY_DIR}/lib)

set(CONFIG_FILE_PATH lib/cmake/${PROJECT_NAME})

configure_package_config_file(${PROJECT_SOURCE_DIR}/${PROJECT_NAME}Config.cmake.in 
	${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
	INSTALL_DESTINATION ${CONFIG_FILE_PATH}
	NO_CHECK_REQUIRED_COMPONENTS_MACRO)

write_basic_package_version_file(${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
        VERSION ${PROJECT_VERSION}
        COMPATIBILITY SameMajorVersion)

# 库的安装命令
install(TARGETS netcdfcxx_legacy netcdfcxx_legacy_static
		EXPORT ${PROJECT_NAME}Targets
		LIBRARY DESTINATION lib
		ARCHIVE DESTINATION lib)
	install(EXPORT ${PROJECT_NAME}Targets
	        DESTINATION ${CONFIG_FILE_PATH})
	install(FILES
        ${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
        ${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
        DESTINATION ${CONFIG_FILE_PATH})
# 头文件安装命令
install(FILES lib/netcdfcpp.h DESTINATION include/netcdfcxx_legacy)
install(FILES lib/ncvalues.h DESTINATION include/netcdfcxx_legacy)

# 以下部分为例子程序的编译
# 设置可执行文件的输出地址
set(EXECUTABLE_OUTPUT_PATH ${PROJECT_BINARY_DIR}/bin)

# 例子的编译方法
macro(add_sample name)
	# 添加可执行文件 命令行
	add_executable(${name} example/${name}.cpp)
	# 为安装文件添加动态库的搜索地址 在Windows下并没有什么用 直接忽略
	set_target_properties(${name} PROPERTIES INSTALL_RPATH ${CMAKE_INSTALL_PREFIX}/lib)
	# 链接动态库
	target_link_libraries(${name} PUBLIC netcdfcxx_legacy)
endmacro()

add_sample(pres_temp_4D_rd)
add_sample(pres_temp_4D_wr)
add_sample(sfc_pres_temp_rd)
add_sample(sfc_pres_temp_wr)
add_sample(simple_xy_rd)
add_sample(simple_xy_wr)
